Laboratorio 2:

Lógica secuencial

lectura de señales asíncronas y módulos genéricos

Diseño automático de sistemas

José Manuel Mendías Cuadros
Dpto. Arquitectura de Computadores y Automática
Universidad Complutense de Madrid

Presentación

- Diseñar un cronómetro de vueltas que cuente segundos módulo 60:
  - Cambio de estado (contando o parado) a activación de startStop.
  - Reinicio síncrono a activación de clear.
  - Parada de visualización (pero continúa la cuenta interna) a activación de lap.
- Leerá las señales y visualizará la cuenta del siguiente modo:
  - Los segundos los mostrará en BCD en los 8 leds menos significativos.
  - El led más significativo parpadeará a 5 Hz.
  - Las señales de entrada las leerá de 3 pulsadores:
    - (startStop, lap, clear) = (btnL, btnC, btnU)
  - Para la medida del tiempo, tomará como base el reloj a 100 MHz.



J.M. Mendías
2024

Entrada elemental

lectura secuencial de pulsadores/interruptores

- La señal digital generada por un pulsador/interruptor tiene las siguientes características:
  - Es asíncrona: puede cambiar en cualquier momento con independencia del reloj.
    - Leída por un sistema secuencial puede llevar a sus biestables a metaestabilidad.
  - Tiene rebotes: cada pulsación se traduce en un tren de pulsos de presión y otro tren de pulsos de depresión.
    - Una única pulsación puede erróneamente interpretarse como una serie de ellas.
  - Es de baja frecuencia: en comparación con las señales síncronas del sistema, es una señal que hace cambia con poca frecuencia y se activa durante largos periodos.
    - Leída por un sistema secuencial de alta frecuencia, una única pulsación se traduce en una serie de valores idénticos leídos durante un gran número de ciclos consecutivos.

- Por ello, toda señal leída por un sistema secuencial y procedente de un pulsador/interruptor es necesario adecuarla a través de:
  - Sincronizador: hace que la señal cambie en instantes síncronos.
  - Eliminador de rebotes: elimina los vaivenes transitorios de la señal.
  - Detector de flancos: convierte una señal que se activa durante varios ciclos en una que lo hace durante un solo ciclo.

J.M. Mendías
2024

Señales asíncronas

problema (i)

- Para que biestable disparado por flanco tenga un comportamiento predecible, la entrada debe estar estable en las proximidades del flanco:
  - Como mínimo debe estar estable durante el tiempo de setup (antes del flanco) y durante tiempo de hold (después del flanco).





J.M. Mendías 2024

Señales asíncronas

problema (ii)

- Cuando se viola el tiempo de hold o el de setup, el biestable entra en un estado metaestable caracterizado por:
  - El retardo de propagación del biestable no está acotado.
  - El valor de salida del biestable es impredecible
  - Si un biestable entra en metaestabilidad puede arrastrar al resto de biestables.
    - Mientras no se estabilice, no lo hacen las señales que dependen de él.



J.M. Mendías 2024

Señales asíncronas

problema (iii)

- Las herramientas EDA garantizan que los circuitos internamente tienen un comportamiento predecible:
  - Porque diseñan la lógica combinacional de manera que todas las señales internas se estabilizan fuera del periodo de apertura de todos los biestables.
  - Lo hacen asegurando que los retardos de la lógica combinacional sintetizada satisfacen las ligaduras de retardo máximo y mínimo establecidas en cada camino.





ligadura de retardo máximo:

$$t_{CLK} \geq (t_d^{ff_1} + t_d^{lc} + t_s^{ff_2})$$

ligadura de retardo mínimo:

$$(t_c^{ff_1} + t_c^{lc}) \geq t_h^{ff_2}$$

Señales asíncronas

problema (iv)

- Sin embargo, una señal externa (o proveniente de otro dominio de reloj) puede cambiar en cualquier momento
  - Pudiendo llevar a metaestabilidad a algunos de los biestables que la leen directa o indirectamente.
  - Que lo haga, depende del momento del ciclo en que se produzca cada uno de los cambios asíncronos y del retardo de la lógica combinacional que atraviese la señal.





J.M. Mendías
2024

Señales asíncronas

sincronizador de 2 etapas (i)

- Un sincronizador es un circuito que retrasa los cambios de una señal al inicio de alguno de los ciclos siguientes.
  - De este modo los cambios de la señal sincronizada nunca se producirán durante el periodo de apertura de los biestables que la leen.
  - La señal sincronizada será una señal interna que las herramientas EDA podrán garantizar que tenga una temporización correcta.



J.M. Mendías
2024

Señales asíncronas

sincronizador de 2 etapas (ii)

- Se construye encadenando un número suficiente de flip-flops.
  - El número necesario depende de las condiciones de funcionamiento del sistema.



- El sincronizador evita que los biestables del sistema entren en metaestabilidad
  - Sin embargo, como la señal de entrada es asíncrona, los biestables que forman el sincronizador pueden entrar en metaestabilidad.

J.M. Mendias
2024

Señales asíncronas

sincronizador de 2 etapas (iii)

- Un sincronizador se dice que en un ciclo dado falla:
  - Si propaga metaestabilidad a su salida, en otro caso se dice que acierta.
- Para que un sincronizador de 2 etapas falle, es necesario:
  - Que la señal de entrada cambie durante el periodo de apertura del primer biestable.
  - Que el primer biestable continúe en estado metaestable pasado un ciclo de reloj
    - De manera que el segundo biestable capture la metaestabilidad.



J.M. Mendías 2024

Señales asíncronas

sincronizador de 2 etapas (iv)

- Calculemos analíticamente la tasa de fallos (nunca nula):
  - La probabilidad de que exista una transición en la entrada del sincronizador durante el tiempo de apertura del primer biestable es:

$$p_a = \frac{t_a}{t_{clk}} = t_a \cdot f_{clk}$$

t_a : tiempo de apertura del biestable
t_{clk} : periodo del reloj de muestreo
f_{clk} : frecuencia del reloj de muestreo

  - La probabilidad de que tras dicha transición el biestable continúe en estado metaestable durante un cierto tiempo de espera es:

$$p_{fallo} = t_a \cdot f_{clk} \cdot e^{\frac{-t_w}{\tau}}$$

τ : tiempo de regeneración
t_w : tiempo de espera

  - Suponiendo que cualquier transición de la entrada puede provocar fallo de sincronización y que el periodo de espera debe ser de 1 ciclo (para que el segundo biestable capture la metaestabilidad del primero):

$$f_{fallo} = f_x \cdot t_a \cdot f_{clk} \cdot e^{\frac{-t_{clk}}{\tau}}$$

f_x : frecuencia de conmutación de la entrada

  - El tiempo medio entre fallos:

$$t_{MTBE} = \frac{1}{f_{fallo}} = \frac{e^{\frac{t_{clk}}{\tau}}}{f_x \cdot t_a \cdot f_{clk}} = \frac{e^{\frac{1}{\tau \cdot f_{clk}}}}{f_x \cdot t_a \cdot f_{clk}}$$

J.M. Mendias 2024

Señales asíncronas

sincronizador de 2 etapas (v)

- Haciendo las siguientes suposiciones:
  - t_a = 2 ns = $2 \cdot 10^{-9}$ s
  - τ = 0.2 ns = $2 \cdot 10^{-10}$ s
- Un sincronizador de 2 etapas que conecta un pulsador a un sistema sistema secuencial a 100 MHz fallará:
  - f_{clk} = 100 MHz = $100 \cdot 10^{6}$ s^{-1}
  - f_x = 100 Hz = $1 \cdot 10^{2}$ s^{-1}
  - t_{MTTF} = $8.22 \cdot 10^{12}$ años (la edad del universo = $13.7 \cdot 10^{9}$ años)
- Sin embargo si aumentamos la frecuencia de reloj a 1 GHz fallará:
  - f_{clk} = 1 GHz = $1 \cdot 10^{9}$ s^{-1}
  - f_x = 100 Hz = $1 \cdot 10^{2}$ s^{-1}
  - t_{MTTF} = 0.7 s

No obstante los fabricantes tratan de reducir t_a y en especial τ de modo que el problema aparece solo a altas frecuencias de reloj:

- 90's: t_a = 50 ns, τ = 0.3 ns
- 10's: t_a = 0.1 ns, τ = 0.04 ns (Xilinx)

Señales asíncronas

sincronizador de 2 etapas (vi)

```
library ieee;
use ieee.std_logic_1164.all;

entity synchronizer is
  port (
    clk   : in  std_logic;
    x     : in  std_logic;
    xSync : out std_logic
  );
end synchronizer;

architecture syn of synchronizer is
begin

  process (clk)
   variable aux : std_logic_vector(1 downto 0) := '0';
  begin
    xSync <= aux(1);
    if rising_edge(clk) then
      aux(1) := aux(0);
      aux(0) := x;
    end if;
  end process;

end syn;
```



sería equivalente usando una señal, pero con variable queda más compacto

El valor en reposo de los pulsadores es '0'

J.M. Mendías
2024

Señales asíncronas

synchronizer.vhd

```
library ieee; use ieee.std_logic_1164.all;

entity synchronizer is
  generic (
    STAGES  : in natural;       -- número de biestables del sincronizador
    XPOL    : in std_logic      -- polaridad (valor en reposo) de la señal a sincronizar
  );
  port (
    clk   : in  std_logic;
    x     : in  std_logic;
    xSync : out std_logic
  );
end synchronizer;

architecture syn of synchronizer is
begin
  process (clk)
    variable aux : std_logic_vector(STAGES-1 downto 0) := (others => XPOL);
  begin
    xSync <= aux(STAGES-1);
    if rising_edge(clk) then
      for i in STAGES-1 downto 1 loop
        aux(i) := aux(i-1);
      end loop;
      aux(0) := x;
    end if;
  end process;
end syn;
```

El módulo se generaliza mediante parametrización

el valor inicial de los biestables debe ser el valor en reposo de la señal a sincronizar

J.M. Mendías
2024

Señales mecánicas

problema

- Toda señal proveniente de un contacto mecánico presenta un vaivén transitorio tras cada cambio de estado.
- Un eliminador de rebotes es un circuito que, conocida la duración del rebote, filtra las transiciones que siguen a todo cambio de estado



J.M. Mendías
2024

Señales mecánicas

eliminador de rebotes

- Se construye usando una FSM que cada vez que detecta un evento en la señal, la mantiene estable durante el tiempo de rebote.



J.M. Mendías
2024

Señales mecánicas

eliminador de rebotes

- Se construye usando una FSM que cada vez que detecta un evento en la señal, la mantiene estable durante el tiempo de rebote.



J.M. Mendías
2024

Señales mecánicas

eliminador de rebotes

- Se construye usando una FSM que cada vez que detecta un evento en la señal, la mantiene estable durante el tiempo de rebote.



J.M. Mendías
2024

Señales mecánicas

eliminador de rebotes

- Se construye usando una FSM que cada vez que detecta un evento en la señal, la mantiene estable durante el tiempo de rebote.



J.M. Mendías
2024

Señales mecánicas

eliminador de rebotes

- Se construye usando una FSM que cada vez que detecta un evento en la señal, la mantiene estable durante el tiempo de rebote.



J.M. Mendías 2024

Señales mecánicas

eliminador de rebotes

- Se construye usando una FSM que cada vez que detecta un evento en la señal, la mantiene estable durante el tiempo de rebote.



J.M. Mendías 2024

Señales mecánicas

eliminador de rebotes

- Se construye usando una FSM que cada vez que detecta un evento en la señal, la mantiene estable durante el tiempo de rebote.



J.M. Mendias
2024

Señales mecánicas

debouncer . vhd

- Se generaliza para el modulo para que pueda eliminar señales de polaridad parametrizable.



J.M. Mendías
2024

Señales mecánicas

debouncer.vhd

```
library ieee;
use ieee.std_logic_1164.all;

entity debouncer is
  generic(
    FREQ_KHZ  : natural;   -- frecuencia de operacion en KHz
    BOUNCE_MS : natural;   -- tiempo de rebote en ms
    XPOL      : std_logic  -- valor en reposo de la señal de la que eliminar rebotes
  );
  port (
    clk  : in  std_logic;
    rst  : in  std_logic;
    x    : in  std_logic;
    xdeb : out std_logic
  );
end debouncer;

use work.common.all;

architecture syn of debouncer is

  signal startTimer, timerEnd: std_logic;

begin
  ...
end syn;
```

El módulo se generaliza mediante parametrización

J.M. Mendías
2024

Señales mecánicas

debouncer.vhd

```
timer:
process (clk)
  constant CYCLES : natural := ms2cycles(FREQ_KHZ, BOUNCE_MS)-1;
  variable count  : natural range 0 to CYCLES-1 := 0;
begin
  if count=0 then
    timerEnd <= '1';
  else
    timerEnd <= '0';
  end if;
  if rising_edge(clk) then
    if rst='1' then
      count := 0;
    else
      if startTimer='1' then
        count := CYCLES-1;
      elsif timerEnd='0' then
        count := count - 1;
      end if;
    end if;
  end if;
end process;
```

función de utilidad definida en common.vhd que calcula ciclos = KHz×ms

la cuenta es local al proceso, es de tipo natural por comodidad. La herramienta determinará el número de bits necesarios

usa una variable porque la FSM no necesita conocer la cuenta, solo necesita saber el final de ella

reset síncrono activo a alta

Señales mecánicas

debouncer.vhd

```
fsm:
process (clk, x)
  type states is (
    waitingKeyDown,
    keyDownDebouncing,
    waitingKeyUp,
    KeyUpDebouncing);
  variable state: states;
begin
  xDeb <= XPOL;
  startTimer <= '0';
  case state is
    when waitingKeyDown =>
      if x=not XPOL then
        startTimer <= '1';
      end if;
    when keyDownDebouncing =>
      xDeb <= not XPOL;
    when waitingKeyUp =>
      xDeb <= not XPOL;
      if x=XPOL then
        startTimer <= '1';
      end if;
    when KeyUpDebouncing =>
      null;
  end case;
...
```

Se declaran identificadores de estado

Se usa variable para compactar el código

Valores por defecto de las salidas

Valores de las salidas distintos de los de defecto

Lógica de salida de la FSM

Señales mecánicas

debouncer.vhd

```
...
  if rising_edge(clk) then
    if rst='1' then
      state := waitingKeyDown;
    else
      case state is
        when waitingKeyDown =>
          if x=not XPOL then
            state := keyDownDebouncing;
          end if;
        when keyDownDebouncing =>
          if timerEnd='1' then
            state := waitingKeyUp;
          end if;
        when waitingKeyUp =>
          if x=XPOL then
            state := KeyUpDebouncing;
          end if;
        when KeyUpDebouncing =>
          if timerEnd='1' then
            state := waitingKeyDown;
          end if;
      end case;
    end if;
  end if;
end process;
```

reset síncrono activo a alta

Lógica de transición de estados de la FSM

J.M. Mendías
2024

Señales de baja frecuencia

problema

- Las señales provenientes de la interacción humana cambian a baja frecuencia.
  - Un valor a escala humana se interpreta como múltiples a escala microelectrónica.
- Un detector de flanco es un circuito que transforma señales activas durante muchos ciclos en señales activas durante un único ciclo.



J.M. Mendías
2024

Señales de baja frecuencia

detector de flancos

- Para detectar los flancos, compara el valor de la señal actual con el que tenía hace un ciclo.



tipo Mealy



tipo Moore

J.M. Mendias
2024

Señales de baja frecuencia

edgedetector.vhd

```
library ieee;
use ieee.std_logic_1164.all;

entity edgeDetector is
  generic (
    XPOL : std_logic
  );
  port (
    clk   : in  std_logic;
    x     : in  std_logic;
    xFall : out std_logic;
    xRise : out std_logic
  );
end edgeDetector;

architecture syn of edgeDetector is
begin

  process (clk)
    variable aux : std_logic_vector(1 downto 0) := (others => XPOL);
  begin
    xFall <= not aux(0) and aux(1);
    xRise <= aux(0) and not aux(1);
    if rising_edge(clk) then
      aux(1) := aux(0);
      aux(0) := x;
    end if;
  end process;

end syn;
```



J.M. Mendías
2024

Diseño principal

esquema RTL



J.M. Mendías
2024

Diseño principal

esquema RTL



J.M. Mendías
2024

Diseño principal

esquema RTL



J.M. Mendias
2024

Diseño principal

esquema RTL



Diseño principal

esquema RTL



Diseño principal

esquema RTL



J.M. Mendias
2024

Diseño principal

esquema RTL



Diseño principal

esquema RTL



Contador modular genérico

modcounter.vhd

```
library ieee;
use ieee.std_logic_1164.all;
use work.common.all;

entity modCounter is
  generic
  (
    MAXVAL : natural   -- valor máximo alcanzable
  );
  port
  (
    clk   : in  std_logic;   -- reloj del sistema
    rst   : in  std_logic;   -- reset (puesta a 0) síncrono
    ce    : in  std_logic;   -- capacitación de cuenta
    tc    : out std_logic;   -- fin de cuenta
    count : out std_logic_vector(log2(MAXVAL)-1 downto 0)   -- cuenta
  );
end modCounter;

library ieee;
use ieee.numeric_std.all;

architecture syn of modCounter is

  signal cs : unsigned(count'range) := (others => '0');

begin
  ...
end syn;
```

El módulo se generaliza mediante parametrización

la anchura del puerto es parametrizable y depende del máximo valor alcanzable por el contador
usa una función de utilidad definida en common.vhd para calcularla

el uso de range es más expresivo y compacto que volver a usar el genérico

J.M. Mendías
2024

Contador modular genérico

modcounter.vhd




```
begin

  stateReg:
  process (clk)
  begin
    if rising_edge(clk) then
      if rst='1' then
        cs <= ...;
      elsif ... then
        if ... then
          cs <= ...;
        else
          cs <= ...;
        end if;
      end if;
    end if;
  end process;

  count <= ...;

  tc <=
    '1' when ... else
    '0';

end syn;
```

Diseño principal

lab2.vhd

```
use work.common.all;

architecture syn of lab2 is

  component modCounter
    ...
  end component;

  constant FREQ_KHZ  : natural := 100_000;  -- frecuencia de operacion en KHz
  constant BOUNCE_MS : natural := 50;       -- tiempo de rebote de los pulsadores en ms

  signal lapTFF, startStopTFF : std_logic := '0';
  signal secLowReg  : std_logic_vector(3 downto 0) := (others => '0');
  signal secHighReg : std_logic_vector(2 downto 0) := (others => '0');

  signal clearSync : std_logic;
  signal startStopSync, startStopDeb, startStopRise : std_logic;
  signal lapSync,  lapDeb, lapRise : std_logic;

  signal cycleCntTC, decCntTC, secLowCntTC : std_logic;

  signal decCnt, secLowCnt : std_logic_vector(3 downto 0);
  signal secHighCnt        : std_logic_vector(2 downto 0);

  signal secLowMux, secHighMux : std_logic_vector(3 downto 0);

begin
  ...
end syn;
```

Registros (con valor inicial)

Conexiones (sin valor inicial)

J.M. Mendías
2024

Diseño principal

lab2.vhd

```
begin

  clearSynchronizer : synchronizer
    generic map ( ... );
    port map ( ... );

  startStopSynchronizer : synchronizer
    generic map ( STAGES => 2, XPOL => '0' );
    port map ( ... );
  startStopDebouncer : debouncer
    generic map ( ... )
    port map ( ... );
  startStopEdgeDetector : edgeDetector
    port map ( ..., xFall => open );

  lapSynchronizer : synchronizer
    generic map ( ... );
    port map ( ... );
  lapDebouncer : debouncer
    generic map ( ... )
    port map ( ... );
  lapEdgeDetector : edgeDetector
    port map ( ... );

  ...
```

sincroniza clear

en reposo los pulsadores valen '0'

sincroniza, elimina rebotes y detecta flancos de startStop

sincroniza, elimina rebotes y detecta flancos de lap

Diseño principal

lab2.vhd

```
...
 cycleCounter : modCounter
   generic map ( MAXVAL => ms2cycles(FREQ_KHZ, 100)-1 )
   port map ( ... );
 decCounter : modCounter
   generic map ( MAXVAL => 9 )
   port map ( ... );
 secLowCounter : modCounter
   generic map ( ... )
   port map ( ... );
 secHighCounter : modCounter
   generic map ( ... )
   port map ( ... );
 lapRegisters :
 process (clk)
 begin
 if rising_edge(clk) then
   if clearSync='1' then
     secLowReg  <= ...;
     secHighReg <= ...;
   elsif ... then
     secLowReg  <= ...;
     secHighReg <= ...;
   end if;
 end process;
 ...
```

contador de ciclos en una decima de segundo (1 ms)

contador 0-9 de décimas de segundo

contador 0-9 de unidades de segundo

contador 0-6 de decenas de segundo

reset síncrono activo a alta

registros de lap (almacenan la cuenta actual cuando se pulsa): módulos con funcionalidad análoga se agrupan en un mismo proceso

Diseño principal

lab2.vhd

```
...

  toggleFFs :
  process (clk)
  begin
    if rising_edge(clk) then
      if clearSync='1' then
        startStopTFF <= ...;
        lapTFF       <= ...;
      else
        if ... then
          startStopTFF <= ...;
        end if;
        if ... then
          lapTFF <= ...;
        end if;
      end if;
    end if;
  end process;

  leftConverterMux :
    secHighMux <=  ... when ... else ...;
  rigthConverterMux :
    secLowMux <=  ... when ... else ...;

  leds <= ...

end syn;
```

reset síncrono activo a alta

biestables T para startStop y lap

multiplexan la salida del contador o del registro de lap

visualiza la salida de los multiplexores

J.M. Mendías
2024

Tareas

1. En la carpeta **DAS/source** crear las carpeta **lab2**.
2. Descargar de la Web los ficheros en:
   - **common:** **synchronizer.vhd**, **debouncer.vhd** y **edgeDetector.vhd**
   - **lab2:** **modCounter.vhd**, **lab2.vhd** y **lab2.xdc**
3. Completar **common.vhd** con la declaración de los nuevos componentes reusables (el contador modular no)
4. Completar el código omitido en los ficheros:
   - **modCounter.vhd** y **lab2.vhd**
5. En la carpeta **DAS/projects** crear el proyecto **lab2**.
6. Incluir en el proyecto (sin copiarlos) los siguientes fuentes:
   - **common.vhd**, **synchronizer.vhd**, **debouncer.vhd**, **edgeDetector.vhd**, **modCounter.vhd**, **lab2.vhd** y **lab2.xdc**
7. Sintetizar, implementar y generar el fichero de configuración.
8. Conectar la placa y encenderla.
9. Volcar el fichero de configuración **lab2.bit**

Acerca de *Creative Commons*

- Licencia CC (Creative Commons)

  - Ofrece algunos derechos a terceras personas bajo ciertas condiciones. Este documento tiene establecidas las siguientes:

Reconocimiento (*Attribution*):
En cualquier explotación de la obra autorizada por la licencia hará falta reconocer la autoría.

No comercial (*Non commercial*):
La explotación de la obra queda limitada a usos no comerciales.

Compartir igual (*Share alike*):
La explotación autorizada incluye la creación de obras derivadas siempre que mantengan la misma licencia al ser divulgadas.

Más información: https://creativecommons.org/licenses/by-nc-sa/4.0/